UNIVERSITÉ DE GENÈVE
Sections d'Informatique et de Physique

FACULTÉ DES SCIENCES
Profs. B. Chopard et M. Droz

Outils, modélisation et simulation en calcul numérique – Corrigé série 11

31 mai 2005

Exercice 1.
Soit L^2 le nombre de sites, N le nombre de réalisations indépendantes du processus, alors la fonction de corrélation à implémenter numériquement s'écrit sous forme développée

$$c_m(t_k) = \frac{\frac{1}{N}\sum_{i=1}^{N} \frac{M_i(t_k)}{L^2}\frac{M_i(0)}{L^2} - \frac{1}{N}\sum_{i=1}^{N} \frac{M_i(t_k)}{L^2}\frac{1}{N}\sum_{i=1}^{N} \frac{M_i(0)}{L^2}}{\frac{1}{N}\sum_{i=1}^{N} \frac{M_i^2(0)}{L^4} - \left[\frac{1}{N}\sum_{i=1}^{N} \frac{M_i(0)}{L^2}\right]^2}, \tag{1}$$

pour $k = 0, 1, 2, \ldots$. $M_i(t_k)$ est l'aimantation totale de la réalisation i au temps k :

$$M_i(t_k) = \sum_{j=1}^{N} s_{ji}(t_k), \tag{2}$$

où $s_{ji}(t_k)$ est la valeur du site j au temps k pour la réalisation i. Idéalement, il faudrait prendre la limite $N \to \infty$ pour avoir un résultat sans fluctuations.

Programme Fortran 90 :

```
module fonctions
module fonctions
implicit none
integer, parameter :: L=20 !system size = L^2
integer, parameter :: Nit=6400 !number of iterations
integer, parameter :: Navg=10**8!number of runs for average
integer, parameter :: Nsave=10**5 !saving every Nsave iterations
real, parameter :: beta=0.2 !interac: J\beta. beta_c=0.44069: infinite system
integer, parameter :: dynamics=1 !=1: Metropolis; else: Glauber
character*(*), parameter :: filename = 's11-1.txt'
contains
!----------------------------------------------------------------------
FUNCTION ran2(idum) !Initialize idum with negative integer.
INTEGER idum,IM1,IM2,IMM1,IA1,IA2,IQ1,IQ2,IR1,IR2,NTAB,NDIV
REAL :: ran2,AM,EPS,RNMX
PARAMETER (IM1=2147483563,IM2=2147483399,AM=1./IM1,IMM1=IM1-1,IA1=40014,&
          &IA2=40692,IQ1=53668,IQ2=52774,IR1=12211,IR2=3791,NTAB=32,&
          &NDIV=1+IMM1/NTAB,EPS=1.2e-7,RNMX=1.-EPS)
INTEGER :: idum2,j,k,iv(NTAB),iy
SAVE iv,iy,idum2
DATA idum2/123456789/, iv/NTAB*0/, iy/0/
if(idum.le.0)then
  idum=max(-idum,1)
  idum2=idum
  do j=NTAB+8,1,-1
    k=idum/IQ1
```

```
    idum=IA1*(idum-k*IQ1)-k*IR1
    if(idum.lt.0)idum=idum+IM1
    if(j.le.NTAB)iv(j)=idum
  enddo
  iy=iv(1)
endif
k=idum/IQ1
idum=IA1*(idum-k*IQ1)-k*IR1
if(idum.lt.0)idum=idum+IM1
k=idum2/IQ2
idum2=IA2*(idum2-k*IQ2)-k*IR2
if(idum2.lt.0)idum2=idum2+IM2
j=1+iy/NDIV
iy=iv(j)-idum2
iv(j)=idum
if(iy.lt.1)iy=iy+IMM1
ran2=min(AM*iy,RNMX)
return
end function ran2
!----------------------------------------------------------------------
end module fonctions
program s11ex1
use fonctions
implicit none
integer :: idum,i,j,k,x,y,xp,xm,yp,ym,magnet
real :: prob,dE,L2
integer, dimension(0:L-1,0:L-1) :: spin
double precision, dimension(1:Nit) :: mt,mtm0
double precision :: m0,m0avg,norm
idum = -4856
mt=0.d0
mtm0=0.d0
m0avg=0.d0
do k=1,Navg !average over independent runs
  do i=0,L-1
    do j=0,L-1
      spin(i,j)=2*int(2.*ran2(idum))-1
    end do
  end do
  magnet=sum(spin)
  m0=dble(magnet)
  m0avg=m0avg+m0
  do i=1,Nit
    x = int(L*ran2(idum))
    y = int(L*ran2(idum))
    xp = mod(x+1,L)
    xm = mod(x-1+L,L)
    yp = mod(y+1,L)
```

```
    ym = mod(y-1+L,L)
    dE = real(2*spin(x,y)*(spin(xp,y) + spin(xm,y) + spin(x,yp) + spin(x,ym)))
    if (dynamics /= 1) then !Glauber
      prob = exp(-beta*dE)
      prob = prob/(1.+prob)
    else !Metropolis
      if (exp(-beta*dE) < 1.) then
        prob = exp(-beta*dE)
      else
        prob = 1.
      end if
    end if
    if (ran2(idum) < prob) then
      spin(x,y) = -spin(x,y)
      magnet = magnet + 2*spin(x,y)
    end if
    mtm0(i) = mtm0(i) + magnet*m0
    mt(i) = mt(i) + magnet
  end do
  if (mod(k,Nsave)==0) then
    L2=real(L**2)
    norm=(mtm0(1)-mt(1)*m0avg/dble(k))
    open(unit=1,file=filename)
    do i=1,Nit
      write(1,*) real(i)/L2,(mtm0(i)-mt(i)*m0avg/dble(k))/norm
    end do
    close(1)
  end if
end do
end program s11ex1
```

Si la dépendance d'une fonction $f(x)$ est exponentielle en x, c'est-à-dire $f(x) = a + b\exp(cx)$, alors en prenant le logarithme de $f(x) - a$ en fonction de x on obtient une relation linéaire. En effet, dans ce cas $\ln[f(x) - a] = \ln b + cx$. La Fig. 1 montre que la décroissance des corrélations en fonction du temps est bien exponentielle.



FIG. 1 – Fonction de corrélation normalisée [de sorte à ce que $c_m(0) = 1$] du modèle d'Ising en fonction du temps pour une dynamique de Metropolis. Système de taille $L^2 = 400$, pour 16 pas de temps et 10^8 réalisations indépendantes. La figure intérieure représente $\log c_m(t)$ en fonction du temps, et la droite en trait-tillés est une interpolation linéaire. Une relation linéaire indique donc une décroissance exponentielle. Une interpolation de la décroissance exponentielle $c_m(t) \sim \exp(-t/\tau_m)$ fournit le temps de relaxation $\tau_m = 4.806\ldots$.

Exercice 2.
Le programme Fortran 90 s'obtient aisément du programme de l'exercice 3.ii de la série 10 :

```
module fonctions
implicit none
integer, parameter :: L=64 !system size = L^2. 8,16,32,64
real, parameter :: beta=0.43 !beta_c=0.44069: infinite system.
integer, parameter :: points=1000 !number of points for the average
integer, parameter :: dynamics=1 !Ndin=1: Metropolis; else: Glauber
character*(*), parameter :: filename = 's11-2.txt'
contains
!----------------------------------------------------------------------
FUNCTION ran2(idum) !Initialize idum with negative integer.
INTEGER idum,IM1,IM2,IMM1,IA1,IA2,IQ1,IQ2,IR1,IR2,NTAB,NDIV
REAL :: ran2,AM,EPS,RNMX
PARAMETER (IM1=2147483563,IM2=2147483399,AM=1./IM1,IMM1=IM1-1,IA1=40014,&
          &IA2=40692,IQ1=53668,IQ2=52774,IR1=12211,IR2=3791,NTAB=32,&
          &NDIV=1+IMM1/NTAB,EPS=1.2e-7,RNMX=1.-EPS)
INTEGER :: idum2,j,k,iv(NTAB),iy
SAVE iv,iy,idum2
DATA idum2/123456789/, iv/NTAB*0/, iy/0/
if(idum.le.0)then
  idum=max(-idum,1)
  idum2=idum
  do j=NTAB+8,1,-1
    k=idum/IQ1
    idum=IA1*(idum-k*IQ1)-k*IR1
    if(idum.lt.0)idum=idum+IM1
    if(j.le.NTAB)iv(j)=idum
  enddo
  iy=iv(1)
endif
k=idum/IQ1
idum=IA1*(idum-k*IQ1)-k*IR1
if(idum.lt.0)idum=idum+IM1
k=idum2/IQ2
idum2=IA2*(idum2-k*IQ2)-k*IR2
if(idum2.lt.0)idum2=idum2+IM2
j=1+iy/NDIV
iy=iv(j)-idum2
iv(j)=idum
if(iy.lt.1)iy=iy+IMM1
ran2=min(AM*iy,RNMX)
return
end function ran2
!----------------------------------------------------------------------
end module fonctions
```

```
program s11ex2
use fonctions
implicit none
integer, parameter :: decorrelation = L !#steps for decorrelation of m
integer :: idum,i,j,k,x,y,xp,xm,yp,ym,magnet
real :: prob,dE
integer, dimension(0:L-1,0:L-1) :: spin
double precision :: mt
idum = -4856
spin=1;magnet=sum(spin)
i=0;k=0;mt=0.d0
average: do
  do j=1,L**2
    x = int(L*ran2(idum))
    y = int(L*ran2(idum))
    xp = mod(x+1,L)
    xm = mod(x-1+L,L)
    yp = mod(y+1,L)
    ym = mod(y-1+L,L)
    dE = real(2*spin(x,y)*(spin(xp,y) + spin(xm,y) + spin(x,yp) + spin(x,ym)))
    if (dynamics /= 1) then !Glauber
      prob = exp(-beta*dE)
      prob = prob/(1.+prob)
    else !Metropolis
      if (exp(-beta*dE) < 1.) then
        prob = exp(-beta*dE)
      else
        prob = 1.
      end if
    end if
    if (ran2(idum) < prob) then
      spin(x,y) = -spin(x,y)
      magnet = magnet + 2*spin(x,y)
    end if
  end do
  i=i+1
  if (mod(i,decorrelation)==0) then !uncorrelatet values
    mt=mt+dabs(dble(magnet))
    k=k+1
    if (k==points) exit average
  end if
end do average
open(unit=1,file=filename)
write(1,*) L,beta,mt/(dble(points)*dble(L**2)),points
close(1)
end program s11ex2
```

La Fig. 2 représente des simulations pour $L = 8, 16, 32, 64$ avec les températures $\beta = 0.6$, 0.5, 0.45, 0.43, 0.4, 0.3.



FIG. 2 – Simulations Monte Carlo pour le modèle d'Ising en deux dimensions. Représentation de $mL^{\beta/\nu}$ en fonction de $|T_c - T|L^{1/\nu}/T_c$ dans un diagramme log-log. On a fait usage des paramètres d'interpolation $\nu = 1$, $\beta = 1/8$, et $T_c = (k_B 0.449687\ldots)^{-1}$.

On constate que pour $T < T_c$ les points s'alignent approximativement sur la courbe supérieure, tandis que pour $T > T_c$ les points s'alignent sur la courbe inférieure. Un meilleur alignement est possible à l'aide d'une plus grande statistique, i.e., de plus longues simulations. La Fig. 3, tirée d'une publication, représente le résultat d'une telle étude avec une meilleure statistique.

Le comportement critique du modèle d'Ising est donc bien décrit par notre approche.



FIG. 3 – Simulations Monte Carlo pour le modèle d'Ising en deux dimensions. Représentation de $mL^{\beta/\nu}$ en fonction de $|T_c - T|L^{1/\nu}/T_c$ dans un diagramme log-log. On constate que la valeur du paramètre d'interpolation $\nu = 0.8$ (image supérieure) regroupe moins précisément les données que pour $\nu = 1.0$ (image inférieure).

Exercice 3.
Cet exercice fait appel à de nombreux résultats de la théorie des phénomènes critiques. Il peut donc être nécessaire de consulter un ouvrage/cours sur le sujet pour compléter ce corrigé. Le théorème de fluctuation-dissipation pour la susceptibilité isotherme χ_T donne

$$\chi_T = \frac{1}{k_B T}\left[\left\langle M^2\right\rangle - \langle M\rangle^2\right]. \tag{3}$$

Comme en moyenne temporelle les fluctuations sont telles que $\langle M\rangle = 0$ (et ceci aussi pour $T < T_c$ si on attend suffisamment longtemps : dans tout système de taille finie il existe une probabilité non nulle de renversement de l'aimantation spontanée), on a

$$\left\langle M^2\right\rangle \simeq k_B T \chi_T. \tag{4}$$

En utilisant la définition de la susceptibilité isotherme

$$\chi_T = \frac{1}{k_B T} \left.\frac{\partial \langle |m|\rangle}{\partial h}\right|_{h=0}, \tag{5}$$

où h est le champ extérieur, l'Eq. (4) devient

$$\left\langle M^2\right\rangle \simeq \left.\frac{\partial \langle |m|\rangle}{\partial h}\right|_{h=0}. \tag{6}$$

Nous avons vu au cours que l'aimantation prend la forme d'échelle

$$m(t,h,l) = l^{-d+y_h} m\left(tl^{1/\nu}, hl^{y_h}, 1\right), \tag{7}$$

où d est la dimension, t est la température réduite définie par $t = (T_c - T)/T$, ν est l'exposant critique de la longueur de corrélation, h est le champ extérieur, $l = N^{1/d}$ est le champ d'échelle de la température, où N est le nombre de sites, et y_h est la dimension anormale du champ d'échelle h. L'Eq. (6) s'écrit donc

$$\left\langle M^2\right\rangle \simeq l^{-d+2y_h} \underbrace{\frac{\partial}{\partial(hl^{y_h})} \langle |m|\rangle \left(tl^{1/\nu}, hl^{y_h}, 1\right)\Big|_{h=0}}_{=\psi_\pm(tl^{1/\nu})}, \tag{8}$$

où ψ_+ décrit le cas $T > T_c$ et ψ_- le cas $T < T_c$. D'autre part, nous avons vu au cours que

$$-d + 2y_h = d - \frac{2\beta}{\nu} = \frac{1}{\nu}(d\nu - 2\beta), \tag{9}$$

ce qui en utilisant la relation de hyperscaling $d\nu = 2 - \alpha$ (α est l'exposant critique de la chaleur spécifique à champ constant, et β est l'exposant critique du paramètre d'ordre – ici l'aimantation spontanée) donne

$$-d + 2y_h = \frac{1}{\nu}(2 - \alpha - 2\beta). \tag{10}$$

En utilisant la relation de scaling $2 - \alpha - 2\beta = \gamma$, où γ est l'exposant critique de la susceptibilité magnétique isotherme, on obtient finalement

$$-d + 2y_h = \frac{\gamma}{\nu}. \tag{11}$$

En insérant l'Eq. (11) dans (8) on a

$$\langle M^2 \rangle = l^{\gamma/\nu} \psi_\pm(t l^{1/\nu}). \tag{12}$$

En utilisant $l = N^{1/d}$ il vient finalement

$$\boxed{\frac{\langle M^2 \rangle}{N} = N^{\gamma/d\nu} \psi_\pm(t N^{1/d\nu}).} \tag{13}$$

Déterminons les formes asymptotiques de $\psi_\pm$. Nous savons que dans la limite thermodynamique (i.e., $N \to \infty$), pour $T > T_c$ et par définition de l'exposant critique γ on a $\chi_T \sim t^{-\gamma}$. Ainsi des Eqs. (13) et (4) il vient

$$\psi_+(N) \sim t^{-\gamma}, \qquad N \to \infty. \tag{14}$$

De même, pour $T < T_c$ par définition de l'exposant β on a $\langle M^2 \rangle \sim t^{2\beta}$. Ainsi des Eqs. (13) et (4) il vient

$$\psi_-(N) \sim t^{2\beta}, \qquad N \to \infty. \tag{15}$$